# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=switchboard-plug-locale
pkgver=2.5.9
pkgrel=1
pkgdesc='Switchboard Locale Plug'
arch=(loong64 x86_64)
url=https://github.com/elementary/switchboard-plug-locale
license=(GPL3)
depends=(
  accountsservice
  glib2
  gnome-desktop
  gtk3
  libgee
  libgranite.so
  libibus
  libswitchboard-2.0.so
  polkit
)
makedepends=(
  git
  meson
  vala
)
_tag=db4ec9703d7ef2a4377c289cb2da93dd3dc1c68a
source=(git+https://github.com/elementary/switchboard-plug-locale.git#tag=${_tag})
sha256sums=('SKIP')

pkgver() {
  cd switchboard-plug-locale

  git describe --tags
}

build() {
  arch-meson switchboard-plug-locale build
  meson compile -C build
}

package() {
  meson install -C build --destdir "${pkgdir}"
}

# vim:set sw=2 sts=-1 et:
